The Physical Bidual Gap and Banach Space Non-Reflexivity:

A Lean 4 Formalization of WLPO via Trace-Class Operators

Paul Chun-Kit Lee*
New York University
dr.paul.c.lee@gmail.com

February 2026

Abstract

We present a Lean 4 formalization of the *Physical Bidual Gap Theorem*, establishing Banach space non-reflexivity of the trace-class operators $S_1(H)$ on a separable Hilbert space and showing that any constructive witness of this non-reflexivity implies the Weak Limited Principle of Omniscience (WLPO). In the algebraic formulation of quantum mechanics, $S_1(H)$ is the natural state space of quantum systems; our result shows that the gap between physical density matrices and the "generalized states" in $(S_1(H))^{**}$ is constructively inaccessible. The formalization comprises 754 lines of Lean 4 code across 8 files, building on Mathlib4 (v4.28.0-rc1), with one interface assumption bridging a companion formalization (Paper 2) for ℓ^1 non-reflexivity—a classical fact (Banach, 1932) whose Lean `axiom` declaration marks an engineering boundary between codebases on different Mathlib4 versions, not a mathematical gap. The backward direction (witness $\Rightarrow$ WLPO) is proven self-contained via an independent Ishihara kernel formalization, providing cross-version verification of Paper 2's core result. The `Classical.choice` dependency in the axiom profile arises from Mathlib4's functional analysis infrastructure; the constructive validity of the equivalence is established by proof-content analysis within the standard CRM methodology (see §6.1). A dependency-free logical skeleton (`P7_Minimal`, 277 lines, 4 files) certifies the reduction chain with no `Classical.choice` in its axiom profile. The key technical contribution is a fully formal proof that every closed subspace of a reflexive Banach space is reflexive, using two applications of the Hahn–Banach theorem.

Contents

1 Introduction **2**

2 Background **3**
2.1 Banach Space Reflexivity 3
2.2 The Bidual Gap Theorem (Paper 2) 4
2.3 Trace-Class Operators 4
2.4 Physical significance 5

3 Mathematical Content **5**

*New York University. AI-assisted formalization; see §7 for methodology.

4 Human-Readable Proofs **6**
4.1 Lemma A: Reflexive Implies Dual Reflexive 6
4.2 Lemma B: Closed Subspace of Reflexive Is Reflexive 6
4.3 Compatibility: Isometry Transfer 7
4.4 Main Theorem: Physical Bidual Gap and Banach Space Non-Reflexivity 7

5 The Lean 4 Formalization **7**
5.1 Architecture 7
5.2 Core Definitions 8
5.3 The Abstract Interface (DiagonalEmbedding.lean) 8
5.4 Lemma A: Reflexive Dual (ReflexiveDual.lean) 9
5.5 Lemma B: Reflexive Subspace (ReflexiveSubspace.lean) 9
5.6 Compatibility (Compat.lean) 10
5.7 Main Assembly (Main.lean) 11
5.8 Key Mathlib4 APIs 11
5.9 Mathlib Version Migration 12
5.10 Reproducibility information 12

6 Interface Assumptions **13**
6.1 The Classical Metatheory 14

7 AI-Assisted Formalization Methodology **14**

8 Discussion and Implications **15**
8.1 Physical interpretation 15
8.2 The Role of WLPO 16
8.3 Formalization as Verification 16
8.4 Open Questions 16

1 Introduction

The Weak Limited Principle of Omniscience (WLPO) is a cornerstone of constructive reverse mathematics. It states:

$$\text{WLPO} \;:\equiv\; \forall \alpha : \mathbb{N} \to \{0,1\},\; \big(\forall n,\; \alpha(n) = 0\big) \;\vee\; \neg\big(\forall n,\; \alpha(n) = 0\big).$$

While classically trivial (an instance of the law of excluded middle), WLPO is constructively independent: it is neither provable nor refutable in Bishop-style constructive mathematics [Bishop, 1967, Bridges and Vîţă, 2006].

In a companion paper [Lee, 2026a], we established the *Bidual Gap Theorem*: for a broad class of Banach spaces, the existence of a constructive witness to non-reflexivity—an element $\Psi \in X^{**} \setminus J_X(X)$—is equivalent to WLPO. This result was formalized in Lean 4 and connected the abstract theory of Banach space duality to a fundamental principle of constructive logic.

The present paper extends that work to a physically motivated setting. In von Neumann's formulation of quantum mechanics, the observables of a quantum system are self-adjoint operators on a separable Hilbert space H, and the states are positive trace-one operators $\rho \in S_1(H)$—the density matrices. The trace norm $\|\rho\|_1 = \mathrm{tr}(|\rho|) = 1$ encodes the normalization of quantum probabilities, and the duality $(S_1(H))^* \cong B(H)$ (bounded operators) is foundational to the algebraic formulation of quantum theory [Bratteli and Robinson, 1987]. Indeed, $S_1(H)$ is the *predual* of $B(H)$: every normal state on the algebra of observables is represented by a density matrix in $S_1(H)$.

The question of Banach space non-reflexivity of $S_1(H)$ is therefore a question about the structure of the quantum state space itself. Non-reflexivity means that the bidual $(S_1(H))^{**}$ is strictly larger than (the canonical image of) $S_1(H)$: there exist "singular states"—elements of $B(H)^*$ that are not representable by any density matrix. Paper 2 [Lee, 2026a] treated the abstract c_0/ℓ^1 setting; the present paper gives the WLPO equivalence its *physical home* in the space of quantum states. We prove:

Theorem 1.1 (Physical Bidual Gap: Banach Space Non-Reflexivity of $S_1(H)$, informal)**.** *Let H be a separable Hilbert space.*

(i) ***(Unconditional)*** *$S_1(H)$ is not reflexive.*

(ii) ***(Constructive bound)*** *Any constructive witness $\Psi \in (S_1(H))^{**} \setminus J_{S_1(H)}(S_1(H))$ implies* WLPO*.*

The first part follows from the chain: c_0 is not reflexive $\Rightarrow$ ℓ^1 is not reflexive $\Rightarrow$ $S_1(H)$ is not reflexive (since ℓ^1 embeds isometrically as a closed subspace of $S_1(H)$ via diagonal embedding). The second part is an instance of the generic result from Paper 2.

The physical interpretation is striking: while $S_1(H)$ provably fails to be reflexive, one cannot *constructively exhibit* a specific element of $(S_1(H))^{**}$ outside the canonical image without invoking WLPO. This connects the structure of quantum state spaces to the foundations of constructive mathematics.

Contributions.

- A Lean 4 formalization of the Physical Bidual Gap Theorem, establishing Banach space non-reflexivity of trace-class operators (754 lines across 8 files, 7 sorry-free, 1 interface assumption for ℓ^1 non-reflexivity bridging Paper 2).
- An independent self-contained formalization of the Ishihara kernel construction (`WLPOFromWitness.lean`, 196 lines, zero custom axioms), proving that any non-reflexivity witness implies WLPO. This provides cross-version verification of the core result from Paper 2, compiled on a different Mathlib4 version.
- A machine-checked proof that every closed subspace of a reflexive Banach space is reflexive (Lemma 3.4), using two applications of the Hahn–Banach theorem—a standard result not currently present in Mathlib4 (Remark 4.2).
- A machine-checked proof that reflexivity transfers across linear isometric equivalences (Lemma 3.5).
- A concrete sorry-backed instance showing $S_1(\ell^2(\mathbb{N}))$ satisfies the abstract trace-class container interface, grounding the theorem in a specific physical space.
- A demonstration of the AI-assisted formalization methodology using Claude Opus 4.6 via the Claude Code CLI.

2 Background

2.1 Banach Space Reflexivity

Let X be a Banach space over $\mathbb{R}$. The *dual space* $X^* = X^*$ is the space of bounded linear functionals $f : X \to \mathbb{R}$, itself a Banach space under the operator norm. The *bidual* $X^{**} = (X^*)^*$ admits a canonical isometric embedding

$$J_X : X \to X^{**}, \qquad J_X(x)(f) = f(x).$$

The space X is *reflexive* if J_X is surjective (equivalently, an isometric isomorphism onto X^{**}).

Classical examples of non-reflexive spaces include c_0, ℓ^1, ℓ^∞, and $L^1(\mu)$ for non-atomic measures. The duality chain

$$c_0 \hookrightarrow (c_0)^* \cong \ell^1 \hookrightarrow (\ell^1)^* \cong \ell^\infty$$

is central to our argument.

2.2 The Bidual Gap Theorem (Paper 2)

Paper 2 [Lee, 2026a] established the following equivalence in Lean 4:

Theorem 2.1 (Bidual Gap Equivalence [Lee, 2026a, Theorem 1])**.**

$$\mathrm{WLPO} \iff \big(\exists \textit{ a separable Banach space } X \textit{ and } \Psi \in X^{**} \setminus J_X(X)\big).$$

Moreover, the forward direction holds uniformly: for any *Banach space* X*, a non-reflexivity witness* $\Psi \in X^{**} \setminus J_X(X)$ *implies* WLPO.

The backward direction ($\Leftarrow$) builds an explicit element $G \in (c_0)^{**}$ that evaluates to 1 on every point-evaluation functional and lies outside $J_{c_0}(c_0)$; the construction is specific to c_0. The forward direction ($\Rightarrow$) constructs an Ishihara kernel from the non-reflexivity witness and extracts WLPO via a constructive consumer; this argument works for any Banach space X and is the direction used by Paper 7.

The proof chain for ℓ^1 not being reflexive proceeds through dual isometries:

1. c_0 is not reflexive (unconditional; the witness G is constructed without WLPO).
2. $(c_0)^* \cong \ell^1$ and $(\ell^1)^* \cong \ell^\infty$ (isometric isomorphisms).
3. If ℓ^1 were reflexive, so would ℓ^∞ be (Lemma A), and then c_0 would be reflexive as a closed subspace of a space isometric to $(\ell^\infty)^*$ (Lemma B), contradicting (1).

2.3 Trace-Class Operators

For a separable Hilbert space H with orthonormal basis $(e_n)_{n\in\mathbb{N}}$, the *trace-class operators* $S_1(H) = S_1(H)$ are the compact operators T on H for which

$$\|T\|_1 \;:=\; \mathrm{tr}(|T|) \;=\; \sum_{n=0}^{\infty} \langle |T| e_n, e_n\rangle < \infty.$$

The trace norm $\|\cdot\|_1$ makes $S_1(H)$ a Banach space, with $(S_1(H))^* \cong B(H)$ (bounded operators) via the pairing $\langle T, A\rangle = \mathrm{tr}(TA)$.

The *diagonal embedding*

$$\iota : \ell^1 \hookrightarrow S_1(H), \qquad \iota(\lambda) = \sum_{n=0}^{\infty} \lambda_n \, |e_n\rangle\langle e_n|$$

is an isometric linear map with closed range: the diagonal operators form a closed subspace of $S_1(H)$ isometric to ℓ^1.

2.4 Physical significance

In algebraic quantum mechanics [Bratteli and Robinson, 1987, 1997], a state on the observable algebra $B(H)$ is a positive normalized linear functional $\omega : B(H) \to \mathbb{R}$. The *normal* states—those that are σ-weakly continuous—are exactly the density matrices in $S_1(H)$, acting via $\omega_\rho(A) = \mathrm{tr}(\rho A)$. The predual relationship $B(H)_* = S_1(H)$ makes the trace-class operators the canonical state space of quantum theory.

Banach space non-reflexivity of $S_1(H)$ means that $(B(H))^* = (S_1(H))^{**} \supsetneq J_{S_1(H)}(S_1(H))$: there exist *singular states* on $B(H)$—positive linear functionals that are not σ-weakly continuous and cannot be represented by any density matrix. These singular states are analogous to finitely additive measures that are not countably additive; they arise from ultrafilter-type constructions and have no direct physical preparation procedure.

The physical importance of this gap extends beyond abstract functional analysis. In quantum statistical mechanics, every state ω on a von Neumann algebra $\mathcal{M}$ admits a unique decomposition $\omega = \omega_n + \omega_s$ into normal and singular parts [Takesaki, 1979]—the noncommutative analogue of the Lebesgue decomposition of measures. The singular component ω_s vanishes on all compact operators: it "sees" only the behavior of observables at infinity.

This decomposition is physically realized in the *thermodynamic limit*. For a quantum system in a finite volume $\Lambda \subset \mathbb{Z}^d$, the equilibrium state at inverse temperature β is a Gibbs state ω_Λ^β—a density matrix, hence a normal state on $B(H_\Lambda)$. In the infinite-volume limit $\Lambda \nearrow \mathbb{Z}^d$, one obtains a state on the quasilocal algebra $\mathfrak{A} = \overline{\bigcup_\Lambda B(H_\Lambda)}$ that may fail to be normal with respect to any fixed representation [Bratteli and Robinson, 1997]. The KMS (Kubo–Martin–Schwinger) states characterizing thermal equilibrium at inverse temperature β are defined by the condition $\omega(AB) = \omega(B\sigma_{i\beta}(A))$ for the modular automorphism group σ_t; at phase transitions, multiple KMS states coexist, corresponding to distinct thermodynamic phases [Haag, 1996].

The passage from finite-volume Gibbs states to infinite-volume KMS states is precisely the passage from the predual $S_1(H)$ to its bidual $(S_1(H))^{**}$. Singular states—elements of the bidual gap—represent idealized thermodynamic configurations that cannot be prepared by any finite laboratory procedure. Their mathematical existence is guaranteed by non-reflexivity; our result shows that *constructively witnessing* any specific singular state requires WLPO.

This connects to a broader question in the foundations of quantum statistical mechanics: what is the logical cost of the thermodynamic limit? Cubitt, Perez-Garcia, and Wolf [Cubitt et al., 2015] showed that the spectral gap problem for quantum many-body systems is *undecidable* at the level of the Halting Problem—far stronger than any omniscience principle. Van Wierst [van Wierst, 2019] explored the paradox of phase transitions from the perspective of constructive mathematics, observing that non-analytic behavior in partition functions (the hallmark of phase transitions) is problematic in frameworks where all total functions are continuous. Our result occupies a specific intermediate position in this landscape: the mere *existence* of the singular sector (non-reflexivity in $\neg$-form) requires WLPO, while the full thermodynamic limit (constructing infinite-volume states via monotone convergence) requires LPO, and non-separable Hahn–Banach separation requires full classical logic.

3 Mathematical Content

We now state the key lemmas and the main theorem precisely.

Definition 3.1 (WLPO)**.** The *Weak Limited Principle of Omniscience* is the proposition

$$\mathrm{WLPO} :\equiv \forall \alpha : \mathbb{N} \to \mathrm{Bool},\ \ (\forall n,\ \alpha(n) = \texttt{false}) \ \vee\ \neg(\forall n,\ \alpha(n) = \texttt{false}).$$

Definition 3.2 (Reflexivity)**.** A normed space X over a field $\mathbb{K}$ is *reflexive* if the canonical embedding $J_X : X \to X^{**}$ (given by $J_X(x)(f) = f(x)$) is surjective.

Lemma 3.3 (Lemma A: Reflexive implies dual reflexive). *If X is reflexive, then X^* is reflexive.*

Lemma 3.4 (Lemma B: Closed subspace of reflexive is reflexive). *Let Y be a closed subspace of a reflexive Banach space X. Then Y is reflexive.*

Lemma 3.5 (Compatibility: Reflexivity transfers across isometries). *If $X \cong Y$ via a linear isometric equivalence and Y is reflexive, then X is reflexive.*

Definition 3.6 (Trace-class container). A *trace-class container* is a separable complete normed space X over $\mathbb{R}$ equipped with an isometric continuous linear map $\iota : \ell^1 \to X$ whose range is closed. The canonical example is $S_1(H)$ with the diagonal embedding.

Theorem 3.7 (Physical Bidual Gap: Banach Space Non-Reflexivity of $S_1(H)$). *Let X be a trace-class container. Then:*

(i) $\neg\,\mathrm{IsReflexive}(\mathbb{R}, X)$.

(ii) $\big(\exists \Psi \in X^{**} \setminus J_X(X)\big) \implies \mathrm{WLPO}$.

4 Human-Readable Proofs

4.1 Lemma A: Reflexive Implies Dual Reflexive

Proof. Let X be reflexive, and let $\varphi \in X^{***} = (X^*)^{**}$. We must find $f \in X^*$ with $J_{X^*}(f) = \varphi$.

Define $f := \varphi \circ J_X \in X^*$, so that for each $x \in X$, $f(x) = \varphi(J_X(x))$.

We verify $J_{X^*}(f) = \varphi$ by checking on all of X^{**}. Since J_X is surjective, any $\Psi \in X^{**}$ has the form $\Psi = J_X(x)$ for some x. Then:

$$J_{X^*}(f)(\Psi) = \Psi(f) = J_X(x)(f) = f(x) = \varphi(J_X(x)) = \varphi(\Psi). \qquad \square$$

4.2 Lemma B: Closed Subspace of Reflexive Is Reflexive

This is the technical bottleneck of the formalization.

Proof. Let $Y \subseteq X$ be a closed subspace with X reflexive, and let $\varphi \in Y^{**}$. We construct $y \in Y$ with $J_Y(y) = \varphi$ in four steps.

Step 1: Lift φ to $\Phi \in X^{**}$. Let $\mathrm{res} : X^* \to Y^*$ be the restriction map $f \mapsto f|_Y$ (formally, $\mathrm{res} = (\cdot) \circ \iota_Y$ where $\iota_Y : Y \hookrightarrow X$ is the inclusion). Define $\Phi := \varphi \circ \mathrm{res} \in X^{**}$.

Step 2: Represent Φ via reflexivity. Since X is reflexive, there exists $x \in X$ with $J_X(x) = \Phi$. That is, for all $f \in X^*$: $f(x) = \Phi(f) = \varphi(f|_Y)$.

Step 3: Show $x \in Y$ by contradiction (Hahn–Banach separation). Suppose $x \notin Y$. Since Y is closed and convex (it is a subspace), the geometric Hahn–Banach separation theorem provides $f_0 \in X^*$ and $u \in \mathbb{R}$ with:

$$\forall a \in Y,\ f_0(a) < u \qquad \text{and} \qquad u < f_0(x).$$

Since Y is a subspace, for any $y \in Y$ and $n \in \mathbb{N}$, we have $n \cdot y \in Y$, so $n \cdot f_0(y) < u$. Letting $n \to \infty$ forces $f_0(y) \le 0$. Applying the same argument to $-y \in Y$ gives $f_0(y) \ge 0$. Hence f_0 annihilates Y: $f_0(y) = 0$ for all $y \in Y$.

In particular, $0 = f_0(0) < u$ (since $0 \in Y$). But also:

$$f_0(x) = \Phi(f_0) = \varphi(\mathrm{res}(f_0)) = \varphi(f_0|_Y) = \varphi(0) = 0,$$

since $f_0|_Y = 0$. This gives $u < f_0(x) = 0 < u$, a contradiction.

Step 4: Verify $J_Y(y) = \varphi$ (Hahn–Banach extension). With $x \in Y$ established, set $y := \langle x, \cdot \rangle \in Y$. For any $g \in Y^*$, the Hahn–Banach extension theorem provides $f \in X^*$ with $f|_Y = g$. Then:

$$J_Y(y)(g) = g(y) = f(x) = \Phi(f) = \varphi(\mathrm{res}(f)) = \varphi(g). \qquad \square$$

Remark 4.1 (Avoidance of James's theorem)**.** The classical route to "closed subspace of reflexive is reflexive" typically passes through James's theorem (X is reflexive if and only if every continuous linear functional attains its norm). Our proof avoids James entirely, using Hahn–Banach directly: geometric separation for Step 3 and norm-preserving extension for Step 4. This is noteworthy because James's theorem is constructively problematic—it relies on a characterization that does not hold in Bishop-style constructive mathematics—whereas our argument uses only the Hahn–Banach theorem, which has a constructive formulation for separable spaces.

Remark 4.2 (Contribution to Mathlib4 infrastructure)**.** Lemma 3.4 appears not to be present in Mathlib4 (as of v4.28). The closest result is `IsReflexive.of_split`, which requires the subspace to be *complemented* (a direct summand)—a strictly stronger condition. Our proof, using geometric Hahn–Banach separation and norm-preserving extension, works for arbitrary closed subspaces and may be of independent interest to the Mathlib4 community.

4.3 Compatibility: Isometry Transfer

Proof of Lemma 3.5. Let $e : X \xrightarrow{\sim} Y$ be a linear isometric equivalence with Y reflexive. Given $\Phi \in X^{**}$, define $e^* : Y^* \to X^*$ by $e^*(g) = g \circ e$ (precomposition). Let $\Psi := \Phi \circ e^* \in Y^{**}$. By reflexivity of Y, there exists y with $J_Y(y) = \Psi$. Set $x := e^{-1}(y)$.

For any $f \in X^*$, define $g := f \circ e^{-1} \in Y^*$. Then $g(y) = f(e^{-1}(y)) = f(x)$, and $\Psi(g) = \Phi(e^*(g)) = \Phi(g \circ e) = \Phi(f \circ e^{-1} \circ e) = \Phi(f)$. Since $J_Y(y)(g) = g(y)$, we get $f(x) = g(y) = J_Y(y)(g) = \Psi(g) = \Phi(f)$. Hence $J_X(x) = \Phi$. $\square$

4.4 Main Theorem: Physical Bidual Gap and Banach Space Non-Reflexivity

Proof of Theorem 3.7. **Part (i).** Let X be a trace-class container with isometric embedding $\iota : \ell^1 \hookrightarrow X$ having closed range. The map ι gives a linear isometry $\ell^1 \to X$, and $\mathrm{range}(\iota)$ is a closed subspace of X. By `LinearIsometry.equivRange`, we obtain $e : \ell^1 \cong \mathrm{range}(\iota)$ as a linear isometric equivalence.

Suppose for contradiction that X is reflexive. By Lemma 3.4, $\mathrm{range}(\iota)$ is reflexive. By Lemma 3.5 (applied to e), ℓ^1 is reflexive—contradicting the known non-reflexivity of ℓ^1.

Part (ii). This follows from the Ishihara kernel construction (proven self-contained in `WLPOFromWitness.lean`, adapted from Paper 2): given any non-reflexivity witness Ψ for any Banach space, one constructs an Ishihara kernel and applies the constructive WLPO consumer. $\square$

5 The Lean 4 Formalization

5.1 Architecture

The formalization consists of 754 lines of Lean 4 code across 8 files, organized as follows:

The dependency graph is:

```
Basic.lean
  |-- ReflexiveDual.lean       (Lemma A)
  |-- DiagonalEmbedding.lean   (HasTraceClassContainer)
  |    |-- Instance.lean       (concrete S1 witness)
  |-- Compat.lean              (isometry transfer)
  |-- ReflexiveSubspace.lean   (Lemma B)
```

File	Role	Lines	Status
`Basic.lean`	WLPO & IsReflexive definitions	34	Complete
`ReflexiveDual.lean`	Lemma A	47	Complete
`DiagonalEmbedding.lean`	`HasTraceClassContainer`	45	Complete
`Compat.lean`	Isometry transfer	78	Complete
`ReflexiveSubspace.lean`	Lemma B (bottleneck)	174	Complete
`WLPOFromWitness.lean`	Ishihara kernel (eliminates axiom)	196	Complete
`Instance.lean`	Concrete S_1 witness (sorry-backed)	51	Stub
`Main.lean`	Assembly + main theorem	129	Complete

Table 1: File manifest for the Paper 7 formalization.

```
|-- WLPOFromWitness.lean     (Ishihara kernel proof)
|
Main.lean                    (assembly)
```

5.2 Core Definitions

The definitions in `Basic.lean` encode WLPO and reflexivity:

```
/-- Weak Limited Principle of Omniscience. -/
def WLPO : Prop :=
  forall (a : Nat -> Bool),
    (forall n, a n = false) ||| not (forall n, a n = false)

/-- Reflexivity: surjectivity of the canonical
    embedding J_X : X -> X**. -/
def IsReflexive (K : Type*) [NontriviallyNormedField K]
    (X : Type*) [NormedAddCommGroup X]
    [NormedSpace K X] : Prop :=
  Function.Surjective (inclusionInDoubleDual K X)
```

Listing 1: Core definitions (Basic.lean).

5.3 The Abstract Interface (DiagonalEmbedding.lean)

Rather than formalizing the full Schatten class infrastructure (not available in MATHLIB4, estimated at 1000+ lines), we axiomatize the essential properties via a typeclass:

```
abbrev ell1 : Type := lp (fun _ : Nat => Real) 1

class HasTraceClassContainer where
  X : Type
  [instNAG : NormedAddCommGroup X]
  [instNS : NormedSpace Real X]
  [instCS : CompleteSpace X]
  [instSep : TopologicalSpace.SeparableSpace X]
  -- Isometric embedding ell1 -> X
  i : ell1 ->L[Real] X
  i_isometry : Isometry i
  i_closedRange : IsClosed (Set.range i)
```

Listing 2: Trace-class container interface (DiagonalEmbedding.lean).

This approach isolates the three properties needed from the embedding (ι is continuous linear, isometric, and has closed range) without requiring the construction of $S_1(H)$.

5.4 Lemma A: Reflexive Dual (ReflexiveDual.lean)

The proof is remarkably concise in Lean 4—just 12 lines of tactic proof:

```
theorem reflexive_dual_of_reflexive
    (hX : IsReflexive K X) :
    IsReflexive K (StrongDual K X) := by
  intro phi
  let f : StrongDual K X :=
    phi.comp (inclusionInDoubleDual K X)
  use f
  ext Psi
  obtain <<x, hx>> := hX Psi
  change Psi f = phi Psi
  rw [<- hx]
  rfl
```

Listing 3: Lemma A: X reflexive implies X* reflexive.

The key insight is that after unfolding the canonical embedding, the goal reduces to a definitional equality (`rfl`). This required adapting to a change in Mathlib4 v4.28 where `inclusionInDoubleDual` unfolds to `ContinuousLinearMap.apply`, altering `simp` behavior.

5.5 Lemma B: Reflexive Subspace (ReflexiveSubspace.lean)

This is the technical bottleneck, requiring two applications of the Hahn–Banach theorem from Mathlib4.

Auxiliary lemma. A key step is showing that a continuous linear functional bounded on a subspace must annihilate it:

```
private lemma annihilates_of_bounded_on_subspace
    {Y : Submodule Real X} (f : StrongDual Real X) (u : Real)
    (hfu : forall a in (Y : Set X), f a < u) :
    forall y in (Y : Set X), f y = 0 := by
  intro y hy
  apply le_antisymm
  . -- f(y) <= 0: for large n, n*f(y) < u
    by_contra h; push_neg at h
    obtain <<n, hn>> := exists_nat_gt (u / f y)
    have hfy_pos : (0 : Real) < f y := h
    have hnu : (n : Real) * f y > u := by
      rwa [gt_iff_lt, <- div_lt_iff_0 hfy_pos]
    have hny : (n : Real) . y in (Y : Set X) :=
      Y.smul_mem (n : Real) hy
    have h_bound := hfu _ hny
    rw [map_smul, smul_eq_mul] at h_bound
    linarith
  . -- f(y) >= 0: similarly via -y
    ...
```

Listing 4: Annihilation lemma for subspace separation.

Main proof body. The four-step proof is shown below (Steps 1–2 and 3–4, with Step 4 nested inside a `suffices`). This is the core of the formalization and, independently of the WLPO story, constitutes a machine-checked proof of a standard functional analysis theorem using two applications of the Hahn–Banach theorem.

```
theorem reflexive_closedSubspace_of_reflexive
    (Y : Submodule Real X) (hYc : IsClosed (Y : Set X))
    (hX : IsReflexive Real X) : IsReflexive Real Y := by
  intro phi
  -- Step 1: Lift phi to X** via restriction
  let res : StrongDual Real X ->L[Real] StrongDual Real Y :=
    ContinuousLinearMap.precomp Real Y.subtypeL
  let Phi : StrongDual Real (StrongDual Real X) :=
    phi.comp res
  -- Step 2: Represent Phi via reflexivity
  obtain <<x, hx>> := hX Phi
  -- Steps 3+4 nested in suffices
  suffices hx_mem : x in (Y : Set X) by
    -- Step 4: Verify J_Y(y) = phi via Hahn-Banach extension
    let y : Y := <<x, hx_mem>>
    use y; ext g
    obtain <<f, hf_ext, _>> :=
      Real.exists_extension_norm_eq Y g
    ...  -- chain: g(y) = f(x) = Phi(f) = phi(g)
  -- Step 3: x in Y by contradiction (Hahn-Banach separation)
  by_contra hx_not_mem
  obtain <<f0, u, hfu, hu_x>> :=
    geometric_hahn_banach_closed_point
      (Y.convex) hYc hx_not_mem
  have h_ann := annihilates_of_bounded_on_subspace f0 u hfu
  have h_res_zero : res f0 = 0 := by
    ext <<z, hz>>
    simp only [res, ContinuousLinearMap.precomp_apply,
      ContinuousLinearMap.comp_apply,
      Submodule.subtypeL_apply,
      ContinuousLinearMap.zero_apply]
    exact h_ann z hz
  -- f0(x) = Phi(f0) = phi(res f0) = phi(0) = 0
  -- but u < f0(x) and 0 < u -- contradiction
  rw [h_res_zero, map_zero] at hf0x
  linarith
```

Listing 5: Lemma B: closed subspace of reflexive is reflexive (main body, ReflexiveSubspace.lean).

The key MATHLIB4 APIs are: `ContinuousLinearMap.precomp` (restriction, Step 1), `geometric_hahn_banac` (separation, Step 3), and `Real.exists_extension_norm_eq` (extension, Step 4).

5.6 Compatibility (Compat.lean)

The isometry transfer proof constructs dual maps via precomposition. A notable technical challenge was the coercion chain in MATHLIB4 v4.28: the path from `LinearIsometryEquiv` to `ContinuousLinearMap` requires passing through `toContinuousLinearEquiv`:

```
let eL : X ->L[K] Y :=
  e.toContinuousLinearEquiv.toContinuousLinearMap
let eLs : Y ->L[K] X :=
  e.symm.toContinuousLinearEquiv.toContinuousLinearMap
let e_star : StrongDual K Y ->L[K] StrongDual K X :=
  ContinuousLinearMap.precomp K eL
```

Listing 6: Coercion chain for isometry transfer.

The critical simp lemmas for this coercion chain are `ContinuousLinearEquiv.coe_coe` and `LinearIsometryEquiv.coe_toContinuousLinearEquiv`.

5.7 Main Assembly (Main.lean)

The forward direction builds a `LinearIsometry` from the typeclass data, converts it to an equivalence via `LinearIsometry.equivRange`, and applies the lemmas in sequence:

```
theorem not_reflexive_of_contains_ell1
    [tc : HasTraceClassContainer] :
    not (IsReflexive Real tc.X) := by
  intro hX
  -- Build LinearIsometry from typeclass data
  let i_li : ell1 ->li[Real] tc.X :=
    { tc.i.toLinearMap with
      norm_map' := fun x =>
        (AddMonoidHomClass.isometry_iff_norm tc.i).mp
          tc.i_isometry x }
  -- LinearIsometryEquiv: ell1 ~ range(i)
  let e := LinearIsometry.equivRange i_li
  let Y : Submodule Real tc.X :=
    LinearMap.range i_li.toLinearMap
  -- Show range(i_li) is closed
  have hYc : IsClosed (Y : Set tc.X) := by
    suffices h : (Y : Set tc.X) = Set.range tc.i by
      rw [h]; exact tc.i_closedRange
    ext x
    simp only [Y, LinearMap.mem_range,
      SetLike.mem_coe, Set.mem_range]
    constructor
    . rintro <<a, ha>>; exact <<a, ha>>
    . rintro <<a, ha>>; exact <<a, ha>>
  -- Closed subspace of reflexive -> reflexive (Lemma B)
  have hY_refl : IsReflexive Real Y :=
    reflexive_closedSubspace_of_reflexive Y hYc hX
  -- Transfer via isometry (Compat)
  have h_ell1_refl : IsReflexive Real ell1 :=
    reflexive_of_linearIsometryEquiv e hY_refl
  -- Contradiction
  exact absurd h_ell1_refl ell1_not_reflexive
```

Listing 7: Forward direction of the main theorem (Main.lean, complete proof).

The combined theorem is a simple conjunction:

```
theorem physical_bidual_gap
    [tc : HasTraceClassContainer] :
    (not (IsReflexive Real tc.X)) /\
    ((exists Psi : StrongDual Real (StrongDual Real tc.X),
        Psi not_in Set.range
          (inclusionInDoubleDual Real tc.X))
      -> WLPO) :=
  <<not_reflexive_of_contains_ell1,
   wlpo_of_traceClass_witness>>
```

Listing 8: The Physical Bidual Gap theorem statement.

5.8 Key Mathlib4 APIs

Table 2 summarizes the Mathlib4 APIs that were essential to the formalization.

API	Purpose
`ContinuousLinearMap.precomp`	Restriction map $X^* \to Y^*$
`Submodule.subtypeL`	Canonical inclusion $Y \hookrightarrow X$
`geometric_hahn_banach_closed_point`	Separation for closed convex sets
`Real.exists_extension_norm_eq`	Norm-preserving Hahn–Banach extension
`LinearIsometry.equivRange`	Isometric equivalence onto range
`ContinuousLinearEquiv.coe_coe`	Coercion chain for equivalences
`inclusionInDoubleDual`	Canonical embedding $J_X : X \to X^{**}$

Table 2: Key Mathlib4 APIs used in the formalization.

5.9 Mathlib Version Migration

During development, the Mathlib4 version was upgraded from v4.23 to v4.28.0-rc1. This required several adaptations:

- **Import paths:** `NormedSpace.HahnBanach.Extension` → `Normed.Module.HahnBanach`; `NormedSpace.HahnBanach.Separation` → `LocallyConvex.Separation`.
- **Lemma renames:** `div_lt_iff` → `div_lt_iff₀`.
- **Coercion changes:** `LinearIsometryEquiv.toContinuousLinearMap` now requires `FiniteDimensional`; the correct path is through `toContinuousLinearEquiv.toContinuousLinearMap`.
- **Definitional unfolding:** `inclusionInDoubleDual` now unfolds to `ContinuousLinearMap.apply`, changing `simp` behavior and requiring explicit `change` tactics.
- **Rewriting vs. simplification:** In the auxiliary annihilation lemma, `simp` with `map_smul, map_neg, smul_eq_mul` was replaced by explicit `rw` steps to produce terms that `linarith` could process.

5.10 Reproducibility information

Reproducibility Box

- **Repository**: `https://github.com/quantmann/FoundationRelativity`
- **LaTeX source & PDF**: `https://doi.org/10.5281/zenodo.18527559`
- **Lean toolchain**: `leanprover/lean4:v4.28.0-rc1`
- **Lake**: `Lake 5.0.0-src+3b0f286 (Lean 4.28.0-rc1)`
- **mathlib4 commit**: `9543d5047cb12a05abd2d9b9bc2ea2a604b3be87`
- **Current commit**: `d7e222386c4a8e9f230daa97fe5e9170ae307207`
- **Build**: `lake exe cache get && lake build`
- **Bundle target**: `Papers` (imports `Main` + `Instance`)
- **Status**: 1 interface assumption (`ell1_not_reflexive`), 1 sorry-backed instance (`Instance.lean`), all 8 files compile (3086 jobs). Axiom profile of `physical_bidual_gap`: `[propext, Classical.choice, Quot.sound, ell1_not_reflexive]`. `wlpo_of_nonreflexive_witness_proof` has no custom axioms.

- **Minimal artifact**: `Papers.P7_PhysicalBidualGap.P7_Minimal.Main` — dependency-free logical skeleton (277 lines, 4 files), no Mathlib4 imports, zero `sorry`, zero errors. Axiomatizes Lemma B and the Ishihara kernel; certifies the reduction chain `NonReflexiveWitness(S₁(H))` $\leftrightarrow$ `WLPO` with no `Classical.choice` in the axiom profile. Build: `lake build Papers.P7_PhysicalBidualGap.P7_Minimal.Main` (5 jobs, 0 errors). `#print axioms` output: `[ell1_closed_subspace_of_S1, ishihara_kernel, not_reflexive_implies_witness_S1, paper2_reverse, reflexive_closedSubspace_of_reflexiv witness_implies_not_reflexive]`.

6 Interface Assumptions

The formalization uses exactly one interface assumption, declared using Lean's `axiom` keyword. The underlying mathematical fact—that ℓ^1 is not reflexive—has been known since Banach's 1932 monograph and is not in question. The Lean `axiom` keyword here marks an *engineering boundary* between two verified codebases on incompatible Mathlib4 versions, not a mathematical gap. A second interface assumption from an earlier version has been eliminated by independently proving the Ishihara kernel construction within Paper 7.

Remaining interface assumption.

1. `ell1_not_reflexive`: $\neg\,\mathrm{IsReflexive}(\mathbb{R}, \ell^1)$.

   Proven in Paper 2 [Lee, 2026a] via the chain: c_0 is not reflexive (unconditional; the witness $G \in (c_0)^{**}$ is constructed without WLPO) $\to$ dual isometries $(c_0)^* \cong \ell^1$ and $(\ell^1)^* \cong \ell^\infty$ $\to$ if ℓ^1 were reflexive, c_0 would be reflexive (via Lemmas A and B), contradiction. Paper 2's WLPO $\leftrightarrow$ bidual gap equivalence is mechanically certified via the dependency-free `P2_Minimal` artifact (zero `sorry`, no `Classical.choice`); the present interface assumption encapsulates the dual-isometry infrastructure of `P2_Full` ($\sim$1,593 lines in `DualIsometriesComplete.lean`).

   Why it remains: This is a version-migration cost, not a logical gap. Paper 2 was developed on Lean v4.23.0-rc2; Paper 7 uses v4.28.0-rc1. The five-version toolchain gap prevents a direct Lake dependency import, and porting Paper 2's dual isometry infrastructure would require migrating $\sim$3,100 lines across 11 files through substantial Mathlib4 API changes. The proposition itself is proved in Paper 2's `P2_Full` target on Lean v4.23.0-rc2; porting to Paper 7's Mathlib4 v4.28 is blocked by version migration cost, not by mathematical content. A unified monorepo on a shared toolchain would eliminate this boundary entirely.

Eliminated interface assumption. The previous `wlpo_of_nonreflexive_witness` axiom—stating that for any Banach space X, a non-reflexivity witness $\Psi \in X^{**} \setminus J_X(X)$ implies WLPO—has been *fully proven* in `WLPOFromWitness.lean` (196 lines). The proof independently formalizes the Ishihara kernel construction from Paper 2 [Lee, 2026a]: given $\Psi \notin \mathrm{range}(J_X)$, we find $h^\star \in X^*$ with $\|\Psi\|/2 < |\Psi(h^\star)|$, define $g(\alpha) :=$ if $(\forall n,\ \alpha_n = 0)$ then 0 else $h^\star$, and set $\delta := |\Psi(h^\star)|/2 > 0$. The resulting Ishihara kernel $(\Psi, 0, g, \delta)$ is consumed by a purely constructive decision procedure to produce WLPO. The axiom profile of `wlpo_of_nonreflexive_witness_proof` contains only Lean's three foundational axioms (`propext`, `Classical.choice`, `Quot.sound`)—no custom axioms.

This constitutes a second machine-checked proof of the hardest direction of the bidual gap equivalence (non-reflexivity witness $\Rightarrow$ WLPO), developed on Mathlib4 v4.28 independently of Paper 2's proof on Mathlib4 v4.23. Two independent formalizations of the same mathematical content compiling cleanly on different library versions provides strong evidence of correctness.

Concrete witness. A sorry-backed `Instance.lean` (51 lines) demonstrates that $S_1(\ell^2(\mathbb{N}))$ satisfies `HasTraceClassContainer`, making the main theorem applicable to a specific physical space. The sorry placeholders correspond to standard Schatten class theory (trace norm isometry and closed range of the diagonal embedding), which is not yet formalized in MATHLIB4.

Conservation. The remaining interface assumption is *conservative*: it asserts a proposition proven in a verified companion formalization. The combined system (Paper 2 + Paper 7) would have zero custom axioms beyond Lean's core type theory and MATHLIB4's standard axioms.

6.1 The Classical Metatheory

Axiom profile. All theorems in the `P7_Full` formalization carry the axiom profile `[propext, Classical.choice, Quot.sound]` plus the interface assumption `ell1_not_reflexive`. The `Classical.choice` dependency enters through MATHLIB4's normed space and functional analysis infrastructure—specifically through the Hahn–Banach theorem (`exists_extension_norm_eq`), the `NormedSpace` typeclass hierarchy, and `Decidable` instances on $\mathbb{R}$.

What the formalization certifies. The Lean formalization provides two forms of evidence. First, *proof correctness*: the theorem statements are correctly formalized, the proofs compile without `sorry` (in 7 of 8 files), and the proof chain—from the Ishihara kernel construction through Lemma B to the main equivalence—is machine-checked. Second, *proof structure*: the forward direction (non-reflexivity witness $\Rightarrow$ WLPO) uses the Ishihara kernel construction, which is a constructive algorithm with no classical case analysis; the reverse direction (WLPO $\Rightarrow$ non-reflexivity) uses WLPO explicitly as a hypothesis; and the Lemma B step (closed subspace of reflexive is reflexive) uses Hahn–Banach extension and geometric separation, both of which are constructively available for separable spaces [Bridges and Vîţă, 2006].

Certification levels. The BISH claim for the overall equivalence rests on mathematical argument about proof content, not on a `Classical.choice`-free axiom certificate from the main formalization. A supplementary artifact (`P7_Minimal`, 277 lines, 4 files) provides a dependency-free logical skeleton—analogous to the `P2_Minimal` artifact of Paper 2 [Lee, 2026a]—that certifies the reduction structure without MATHLIB4. In `P7_Minimal`, the forward direction (Ishihara kernel) and Lemma B are axiomatized with references to their `P7_Full` proofs, and the logical chain is verified to use no classical axioms beyond the explicitly listed assumptions: the `#print axioms` output for `P7_main` contains only the six declared interface axioms—notably absent are `Classical.choice`, `propext`, and `Quot.sound`. The `P7_Full` proofs supply the mathematical content; `P7_Minimal` certifies the logical architecture.

A systematic treatment of the relationship between MATHLIB4's classical foundations and the constructive claims across the paper series is given in Paper 10 [Lee, 2026b, forthcoming], which establishes three certification levels—*mechanically certified*, *structurally verified*, and *paper-level*—and classifies each paper accordingly. Paper 7 is classified as "structurally verified" (upgraded from "paper-level" upon completion of `P7_Minimal`).

7 AI-Assisted Formalization Methodology

This formalization was developed using **Claude Opus 4.6** (Anthropic, 2026) via the **Claude Code** command-line interface, following the same human–AI workflow as Paper 2 [Lee, 2026a]. The human author wrote a mathematical blueprint (`PhysicalBidualGap_Blueprint.md`, 520 lines) specifying all theorem statements, proof strategies, and target MATHLIB4 APIs. Claude Opus 4.6 then explored the MATHLIB4 codebase to locate exact API signatures and import paths, generated the Lean 4 proof terms, and handled the MATHLIB4 version migration from v4.23 to v4.28 (adapting to renamed lemmas, changed coercion behavior, and deprecated import paths). The human author reviewed all proofs for mathematical correctness and Mathlib4 conventions. Final verification was by `lake build` (0 errors, 0 warnings, 0 `sorry`).

Task	**Human**	**AI (Claude Opus 4.6)**
Mathematical blueprint	✓	
Proof strategy design	✓	
Mathlib4 API discovery		✓
Lean proof generation		✓
Proof review	✓	
Version migration		✓
Build verification		✓
Paper writing	✓	✓

Table 3: Division of labor between human and AI.

Comparison with Paper 2. The methodology closely follows Paper 2, but Paper 7 required more delicate interaction with Mathlib4's Hahn–Banach infrastructure and the additional challenge of a Mathlib4 version migration (v4.23 → v4.28) that broke coercion chains and renamed lemmas across multiple files.

8 Discussion and Implications

8.1 Physical interpretation

The Physical Bidual Gap Theorem reveals a constructive limitation in the theory of quantum state spaces. While Banach space non-reflexivity of $S_1(H)$ is provable without WLPO—meaning the bidual $(S_1(H))^{**}$ is strictly larger than (the image of) $S_1(H)$—one cannot *constructively exhibit* any specific element of the gap without WLPO.

In the algebraic formulation of quantum mechanics, the bidual $(S_1(H))^{**}$ contains "singular states" on $B(H)$: positive linear functionals that are not σ-weakly continuous and cannot be represented by any density matrix. Classically, Gleason's theorem guarantees that in dimension ≥ 3, every countably additive probability measure on the projection lattice of H extends to a normal state—but our result concerns the *finitely* additive measures, the singular states that live in the bidual gap. Their existence is provable, but constructing any specific one requires WLPO.

The thermodynamic limit. The physical home of singular states is the thermodynamic limit of quantum statistical mechanics. In the Haag–Kastler framework for algebraic quantum field theory [Haag, 1996], observables localized in bounded spacetime regions generate a net of local algebras; the quasilocal algebra is their norm closure. States on this algebra decompose into *folia*—equivalence classes under quasi-equivalence of GNS representations [Bratteli and Robinson, 1987]. Normal states belong to the folium of the identity representation; singular states lie outside every folium. Via the GNS construction, singular states give rise to representations *disjoint* from the identity representation—physically, they correspond to inequivalent thermodynamic phases or superselection sectors.

The decomposition $\omega = \omega_n + \omega_s$ into normal and singular parts [Takesaki, 1979] is the noncommutative Lebesgue decomposition. Our theorem calibrates the logical cost of witnessing the singular part: determining whether $\omega_s \neq 0$ for a constructively given state is a zero-test on $\|\omega_s\|$, and zero-tests for reals are precisely what WLPO governs.

Calibration landscape. Our result fits into a broader stratification of the logical costs of quantum statistical mechanics over constructive mathematics (BISH):

Physical statement	**Minimal principle**
Finite-volume Gibbs state properties	BISH (constructive)
$S_1(H)$ is not reflexive ($\neg$-form)	WLPO
Singular state witness ($\exists\, \Psi \in (S_1(H))^{**} \setminus J_{S_1(H)}(S_1(H))$)	$\geq$ WLPO
Thermodynamic limit via monotone convergence	LPO
Hahn–Banach separation in non-separable duals	LEM
Spectral gap decidability (general)	Undecidable [Cubitt et al., 2015]

The first row is fully constructive: Peierls-type bounds, finite-volume magnetization estimates, and explicit Gibbs state calculations require no omniscience principles. Our theorem occupies the second and third rows. The fourth row reflects the observation that constructing infinite-volume limits via monotone convergence of partition functions requires the Limited Principle of Omniscience (LPO), which is strictly stronger than WLPO [Bridges and Vîţă, 2006]. The fifth row marks the classical boundary: full Hahn–Banach separation in non-separable duals (as needed to construct singular functionals on ℓ^∞ directly) requires the law of excluded middle. The sixth row, due to Cubitt, Perez-Garcia, and Wolf [Cubitt et al., 2015], places general spectral gap determination at the level of the Halting Problem—far beyond any omniscience principle.

This landscape suggests that WLPO occupies a natural position in the foundations of quantum theory: it is the precise logical cost of the *ontological status* of the singular sector—the assertion that singular states exist as concrete mathematical objects, rather than merely as a logical impossibility of reflexivity.

8.2 The Role of WLPO

WLPO sits at a precise level in the constructive hierarchy: it is implied by the Limited Principle of Omniscience (LPO), which also independently implies Markov's Principle (MP), and it implies the Lesser Limited Principle of Omniscience (LLPO). Thus LPO $\Rightarrow$ WLPO $\Rightarrow$ LLPO, while MP is independent of WLPO over BISH—neither implies the other [Bridges and Vîţă, 2006]. Our result places the Banach space non-reflexivity witness for $S_1(H)$ at this specific level of the constructive hierarchy, rather than requiring full classical logic.

8.3 Formalization as Verification

The machine-checked nature of the proof provides a high degree of confidence in the result. The single remaining interface assumption is clearly delineated and corresponds to a proven result in a companion formalization. The proof structure—definition of the abstract interface, modular lemmas, and clean assembly—demonstrates that substantial functional analysis can be formalized in Lean 4 with existing Mathlib4 infrastructure.

8.4 Open Questions

1. **Eliminating the remaining axiom:** The sole remaining interface assumption (`ell1_not_reflexive`) requires porting Paper 2's dual isometry chain (∼1,593 lines) to Lean v4.28. A unified monorepo with a shared Mathlib pin would eliminate this bridge entirely.

2. **Schatten classes in Mathlib4:** A formalization of the Schatten p-classes $S_p(H)$ would replace both the abstract `HasTraceClassContainer` and the sorry-backed `Instance.lean` with concrete constructions, strengthening the result.

3. **Generalization:** Does the equivalence extend to other physically relevant non-reflexive spaces, such as the space of bounded operators $B(H)$ or the predual of a von Neumann algebra?

4. **Quantitative gaps:** Can the "size" of the bidual gap (e.g., in terms of cardinality or density character) be characterized constructively?

Acknowledgments

The Lean 4 formalization was developed using Claude Opus 4.6 (Anthropic, 2026) via the Claude Code CLI tool. We thank the MATHLIB4 community for maintaining the comprehensive library of formalized mathematics that made this work possible.

References

E. Bishop. *Foundations of Constructive Analysis.* McGraw-Hill, New York, 1967.

E. Bishop and D. Bridges. *Constructive Analysis.* Grundlehren der mathematischen Wissenschaften, vol. 279. Springer-Verlag, Berlin, 1985.

O. Bratteli and D. W. Robinson. *Operator Algebras and Quantum Statistical Mechanics*, vol. 1. Texts and Monographs in Physics. Springer-Verlag, New York, 2nd edition, 1987.

O. Bratteli and D. W. Robinson. *Operator Algebras and Quantum Statistical Mechanics*, vol. 2: Equilibrium States, Models in Quantum Statistical Mechanics. Texts and Monographs in Physics. Springer-Verlag, Berlin, 2nd edition, 1997.

D. Bridges and L. S. Vîţă. *Techniques of Constructive Analysis.* Universitext. Springer, New York, 2006.

T. S. Cubitt, D. Perez-Garcia, and M. M. Wolf. Undecidability of the spectral gap. *Nature*, 528(7581):207–211, 2015.

L. de Moura, S. Kong, J. Avigad, F. van Doorn, and M. von Raumer. The Lean theorem prover (system description). In *CADE-25*, LNAI 9195, pages 378–388. Springer, 2015. Lean 4: `https://lean-lang.org/`, 2021–present.

R. Haag. *Local Quantum Physics: Fields, Particles, Algebras.* Texts and Monographs in Physics. Springer-Verlag, Berlin, 2nd edition, 1996.

H. Ishihara. Continuity properties in constructive mathematics. *Journal of Symbolic Logic*, 57(2):557–565, 1992.

P. C.-K. Lee. The bidual gap: A Lean 4 formalization of WLPO and non-reflexivity in Banach spaces. Preprint, 2026. Lean 4 formalization: `https://github.com/quantmann/FoundationRelativity`.

P. C.-K. Lee. The physical bidual gap and Banach space non-reflexivity: a Lean 4 formalization of WLPO via trace-class operators. Preprint, 2026. Lean 4 formalization: `https://github.com/quantmann/FoundationRelativity`.

Mathlib Community. *Mathlib*: the math library for Lean. `https://leanprover-community.github.io/mathlib4_docs/`, 2020–present.

M. Takesaki. *Theory of Operator Algebras I.* Springer-Verlag, New York, 1979.

P. van Wierst. The paradox of phase transitions in the light of constructive mathematics. *Synthese*, 196(5):1863–1884, 2019.

Anthropic. Claude Opus 4.6 and Claude Code CLI. `https://www.anthropic.com/claude`, 2026.